
========================DIFF_STAGE_Find_Harmonics====================================

==========Define_Simulation_Spectrum=================================
totalperiod = 0.005   BinBandWidth = 200
sampleRate  = 2.5E-05 nyquist      = 20000
==========Run_Transient==============================================
==========Run_FFT====================================================
==========Define_FFT_Points==========================================
BinsPerHarmonic  = 5
==========Printout_FFT_Points========================================
index =4   freq = 1000  value = 0.000110669  dist_% = 100
index =9   freq = 2000  value = 4.50714E-12  dist_% = 4.07263E-06
index =14  freq = 3000  value = 1.80348E-05  dist_% = 16.2961
index =19  freq = 4000  value = 4.67826E-13  dist_% = 4.22725E-07
index =24  freq = 5000  value = 3.94386E-06  dist_% = 3.56364
index =29  freq = 6000  value = 4.62211E-12  dist_% = 4.17651E-06
index =34  freq = 7000  value = 8.88489E-07  dist_% = 0.802833
index =39  freq = 8000  value = 2.22783E-12  dist_% = 2.01305E-06

DIFF_STAGE_Find_Harmonics

*
.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VC     0      DC     10
VTime     VTime  0      DC     0      PWL(   0      0    1      1)
Vfreq1    Vfreq1 0      DC     1000
BVIN      VIN    0      V      = .1*cos(6.283185307179586*V(VFreq1)*V(VTime))
BOUT      OUT    0      V      = V(VCN1)-V(VCN2)
I1        VEN1   0      100u
QN1       VCN1   VIN    VEN1   NPN1   1.00
QN2       VCN2   0      VEN1   NPN1   1.00
R1        VCN1   VC     1
R2        VCN2   VC     1
*================Model_Files=======================================================
.model    NPN1   NPN(   BF=2100 VAF=216 )
.model    PNP1   PNP(   BF=2100  VAF=21)
.control
echo  "==========Define_Simulation_Spectrum====================================="
let tperiod =    5m
let sampRate =   25u
let BinBW =      1/tperiod
let nyquist =    .5/sampRate
echo            "totalperiod = $&tperiod  BinBandWidth = $&BinBW"
echo            "sampleRate  = $&sampRate nyquist      = $&nyquist"
echo  "==========Run_Transient=============================================="
tran             25u                     5m         0    1u
plot             out
echo  "==========Run_FFT=============================================="
linearize
set              specwindow=   "rectangular"
spec             200    20k      200     v(out)
plot             mag(out) loglog
echo  "==========Define_FFT_Points=============================================="
let period =     5m
let inFreq =     1k
let binBW =      1/period
let FundBin=     inFreq/binBW 
echo            "BinsPerHarmonic  = $&FundBin"
echo  "==========Printout_FFT_Points=============================================="
let i =          0
repeat           8
let i =          i +1



let index =      i*FundBin -1
let findex =     FundBin -1
let freq =       frequency[$&index]
let val =        mag(out[$&index])
let dist =       100*val/mag(out[$&findex])
echo            "index =$&index  freq = $&freq  value = $&val  dist_% = $&dist"
endrepeat
.endc
.end


